Corrigé du TP

201-EEDM (Automne 2004)

Pierre Lantagne

Initialisation

```
> restart;
> with(plottools,disk):
  with(plots,display):
> interface(imaginaryunit=i);
```

$$I \quad \textbf{(1.1)}$$

No. 1

a) Soit $P = k\,x^3 + x^2 + k^2\,x + 3\,k^2 + 11$. Nous devons avoir $P(-2) = 0$ afin que $x+2$ divise le polynôme P. Il faut donc résoudre pour k, l'équation $P(-2) = 0$.

```
> P:=k*x^3+x^2+k^2*x+3*k^2+11;
```

$$P := k\,x^3 + k^2\,x + 3\,k^2 + x^2 + 11 \quad \textbf{(2.1)}$$

```
> Solution:=solve(eval(P,x=-2)=0,{k});
```

$$Solution := \{k=5\}, \{k=3\} \quad \textbf{(2.2)}$$

Si $k=5$ ou $k=3$, $P(x)$ est divisible par $x+2$.

b) Soit $P = x^3 - k\,x^2 - 14\,x + 15\,k$. Puisque P(5) donne le reste de la division du polynôme P par $x-5$, résolvons pour k, l'équation $P(5) = k$.

```
> P:=x^3-k*x^2-14*x+15*k;
```

$$P := -k\,x^2 + x^3 + 15\,k - 14\,x \quad \textbf{(2.3)}$$

```
> solve(eval(P,x=5)=k,{k});
```

$$\{k=5\} \quad \textbf{(2.4)}$$

Si $k=5$, le reste de la division du polynôme P par $(x-5)$ est donc égale à 5.

```
>
```

No. 2

i) Soit $P(x) = x^3 - i\,x^2 + 2\,i\,x + 2$. Sachant que r est une racine de P(x) ssi P(r)=0, on a que

```
> P:=x^3-i*x^2+2*i*x+2;
```

$$P := x^3 - \mathrm{i}\,x^2 + 2\,\mathrm{i}\,x + 2 \quad \textbf{(3.1)}$$

```
> 'P'(i)=eval(P,x=i);
```

$$P(\mathrm{i}) = 0 \quad \textbf{(3.2)}$$

Ce qui montre que i est une racine du polynôme P.

```
> 'P'(-i)=eval(P,x=-i);
```

$$P(-\mathrm{i}) = 4 + 2\,\mathrm{i} \quad \textbf{(3.3)}$$

Ce qui montre que $-i$ n'est pas une racine du polunôme P.

ii) Bien que i soit une racine de P et que $-i$ n'en soit pas une, cela ne contredit en rien le théorème sur les paires de racines conjugués d'un polynôme car la condition d'applicabilité de ce théorème n'est pas satisfaite, c'est-à-dire que le polynôme $P = x^3 - i\,x^2 + 2\,i\,x + 2$ n'est pas un polynôme à coefficients réels.

No. 3

La parabole verticale cherchée est d'équation $y = P = a\,x^2 + b\,x + c$. Résolvons alors le système d'équations suivant.

$$\mathrm{P}(2) = 4\,a + 2\,b + c = \sqrt{2} + 2$$
$$\mathrm{P}(3) = 9\,a + 3\,b + c = 3^{\frac{1}{3}} - 3$$
$$\mathrm{P}(4) = 16\,a + 4\,b + c = 4^{\frac{1}{4}} - 4$$

```
> P:=a*x^2+b*x+c;
```

$$P := a\,x^2 + b\,x + c \tag{4.1}$$

Établissons ce système d'équations.

```
> Éq1:=eval(P,x=2)=sqrt(2)+2;
  Éq2:=eval(P,x=3)=surd(3,3)-3;
  Éq3:=eval(P,x=4)=surd(4,4)-4;
```

$$\begin{gathered} \acute{E}q1 := 4\,a + 2\,b + c = 2 + \sqrt{2} \\ \acute{E}q2 := 9\,a + 3\,b + c = 3^{1/3} - 3 \\ \acute{E}q3 := 16\,a + 4\,b + c = \sqrt{2} - 4 \end{gathered} \tag{4.2}$$

Résolvons ce système.

```
> Solution:=solve({Éq1,Éq2,Éq3},{a,b,c});
```

$$Solution := \left\{a = 2 + \sqrt{2} - 3^{1/3}, b = -15 - 6\sqrt{2} + 6\,3^{1/3}, c = 24 + 9\sqrt{2} - 8\,3^{1/3}\right\} \tag{4.3}$$

```
> assign(%);
> 'P'(x)=P;
```

$$P(x) = \left(2 + \sqrt{2} - 3^{1/3}\right) x^2 + \left(-15 - 6\sqrt{2} + 6\,3^{1/3}\right) x + 24 + 9\sqrt{2} - 8\,3^{1/3} \tag{4.4}$$

Le polynôme cherché est donc

$$\left(2 + \sqrt{2} - 3^{\frac{1}{3}}\right) x^2 + \left(-15 - 6\sqrt{2} + 6\,3^{\frac{1}{3}}\right) x + 24 + 9\sqrt{2} - 8\,3^{\frac{1}{3}}.$$

Illustrons, dans un même graphique, les trois points donnés ainsi que la parabole trouvée.

```
> Point1:=disk([2,sqrt(2)+2],0.1,color=orange):
  Point2:=disk([3,surd(3,3)-3],0.1,color=orange):
  Point3:=disk([4,surd(4,4)-4],0.1,color=orange):
  Parabole:=plot([x,P,x=1..6],color=navy):
> display({Point1,Point2,Point3,Parabole},scaling=constrained,
  size=[300,300],view=[1..6,-3..5],axesfont=[TIMES,ROMAN,8]);
```




```
> unassign('a','b','c');
```

No. 4

Le polynôme à coefficients complexes de degré minimal cherché est de degré trois. En effet, en ne prenant pas en compte ni les conjugués complexes ni les conjugués algébriques des racines données, on obtiendra un polynôme dont les coefficients ne seront pas tous réels et ni tous rationnels.

```
> P:=2*(x-sqrt(5))*(x-(2-i))*(x-2);
```

$$P := 2\left(x-\sqrt{5}\right)(x-2+\mathrm{i})(x-2) \tag{5.1}$$

```
> Polynôme:=collect(%,x);
```

$$Polynôme := 2x^3+\left(-2\sqrt{5}-8+2\,\mathrm{i}\right)x^2+\left((4-2\,\mathrm{i})\sqrt{5}+4\sqrt{5}+8-4\,\mathrm{i}\right)x+(-8+4\,\mathrm{i})\sqrt{5} \tag{5.2}$$

Le polynôme à coefficients complexes de degré minimal possédant les racines $\sqrt{5}$, $2-i$ et 2, ayant le coefficient dominant égal à 2 est donc

$$2x^3+\left(-8+2i-2\sqrt{5}\right)x^2+\left(8-4i-2i\sqrt{5}+8\sqrt{5}\right)x+4i\sqrt{5}-8\sqrt{5}$$

```
> unassign('a','b','c');
```

No. 5

Montrons que le polynôme

$P=\dfrac{a_1\,(x-r_2)\,(x-r_3)}{(r_1-r_2)\,(r_1-r_3)}+\dfrac{a_2\,(x-r_1)\,(x-r_3)}{(r_2-r_1)\,(r_2-r_3)}+\dfrac{a_3\,(x-r_1)\,(x-r_2)}{(r_3-r_1)\,(r_3-r_2)}$ est le seul polynôme quadratique tel que $P(r_1)=a_1$, $P(r_2)=a_2$, $P(r_3)=a_3$.

Vérifions d'abord que P vérifie les égalités $P(r_1)=a_1$, $P(r_2)=a_2$, $P(r_3)=a_3$.

```
> P:= a1*(x-r[2])*(x-r[3])/(r[1]-r[2])/(r[1]-r[3])+a2*(x-r[1])*(x-r[3])/(r[2]-r[1])/(r[2]-r[3])+a3*(x-r[1])*(x-r[2])/(r[3]-r[1])/(r[3]-r[2]);
```

$$P := \frac{a1\,(x-r_2)\,(x-r_3)}{(r_1-r_2)\,(r_1-r_3)} + \frac{a2\,(x-r_1)\,(x-r_3)}{(r_2-r_1)\,(r_2-r_3)} + \frac{a3\,(x-r_1)\,(x-r_2)}{(r_3-r_1)\,(r_3-r_2)} \tag{6.1}$$

```
> Eval(P,x=r[1])=eval(P,x=r[1]);
  Eval(P,x=r[2])=eval(P,x=r[2]);
  Eval(P,x=r[31])=eval(P,x=r[3]);
```

$$\left.\left(\frac{a1\,(x-r_2)\,(x-r_3)}{(r_1-r_2)\,(r_1-r_3)} + \frac{a2\,(x-r_1)\,(x-r_3)}{(r_2-r_1)\,(r_2-r_3)} + \frac{a3\,(x-r_1)\,(x-r_2)}{(r_3-r_1)\,(r_3-r_2)}\right)\right|_{x=r_1} = a1$$

$$\left.\left(\frac{a1\,(x-r_2)\,(x-r_3)}{(r_1-r_2)\,(r_1-r_3)} + \frac{a2\,(x-r_1)\,(x-r_3)}{(r_2-r_1)\,(r_2-r_3)} + \frac{a3\,(x-r_1)\,(x-r_2)}{(r_3-r_1)\,(r_3-r_2)}\right)\right|_{x=r_2} = a2$$

$$\left.\left(\frac{a1\,(x-r_2)\,(x-r_3)}{(r_1-r_2)\,(r_1-r_3)} + \frac{a2\,(x-r_1)\,(x-r_3)}{(r_2-r_1)\,(r_2-r_3)} + \frac{a3\,(x-r_1)\,(x-r_2)}{(r_3-r_1)\,(r_3-r_2)}\right)\right|_{x=r_{31}} = a3 \tag{6.2}$$

Nous allons montrer que le polynône P est le seul en faisant une preuve par contradiction (*reductio ad absurdum*). Dans un premier temps, nous allons supposer l'existence d'un polynôme quadratique Q,différent du polynôme P, respectant lui-aussi les égalités $Q(r_1) = a_1$, $Q(r_2) = a_2$, $Q(r_3) = a_3$.

Soit $Q = a\,x^2 + b\,x + c$ cet autre polynôme quadratique différent de P vérifiant les égalités $Q(r_1) = a_1$, $Q(r_2) = a_2$ et $Q(r_3) = a_3$.

Obtenons les coefficients du polynôme Q en résolvant pour *a*, *b* et *c*, le système{ $Q(r_1) = a_1$, $Q(r_2) = a_2$ et $Q(r_3) = a_3$ }.

```
> Q:=a*x^2+b*x+c;
```

$$Q := a\,x^2 + b\,x + c \tag{6.3}$$

```
> Éq1:=eval(Q,x=r[1])=a1;
  Éq2:=eval(Q,x=r[2])=a2;
  Éq3:=eval(Q,x=r[3])=a3;
```

$$\acute{E}q1 := a\,r_1^2 + b\,r_1 + c = a1$$

$$\acute{E}q2 := a\,r_2^2 + b\,r_2 + c = a2$$

$$\acute{E}q3 := a\,r_3^2 + b\,r_3 + c = a3 \tag{6.4}$$

```
> Solution:=solve({Éq1,Éq2,Éq3},{a,b,c});
```

$$Solution := \left\{a = \frac{a1\,r_2 - a1\,r_3 - a2\,r_1 + a2\,r_3 + a3\,r_1 - a3\,r_2}{r_1^2\,r_2 - r_1^2\,r_3 - r_1\,r_2^2 + r_1\,r_3^2 + r_2^2\,r_3 - r_2\,r_3^2}, b = \right. \tag{6.5}$$

$$-\frac{a1\, r_2^2 - a1\, r_3^2 - a2\, r_1^2 + a2\, r_3^2 + a3\, r_1^2 - a3\, r_2^2}{(r_1 - r_2)\,(r_1 r_2 - r_1 r_3 - r_2 r_3 + r_3^2)}, c = \frac{a1\, r_2^2 r_3 - a1\, r_2 r_3^2 - a2\, r_1^2 r_3 + a2\, r_1 r_3^2 + a3\, r_1^2 r_2 - a3\, r_1 r_2^2}{(r_1 - r_2)\,(r_1 r_2 - r_1 r_3 - r_2 r_3 + r_3^2)}\Bigg\}$$

```
> assign(%);
```

Le polynôme Q est donc le suivant.

```
> Q;
```

$$\frac{(a1\, r_2 - a1\, r_3 - a2\, r_1 + a2\, r_3 + a3\, r_1 - a3\, r_2)\, x^2}{r_1^2 r_2 - r_1^2 r_3 - r_1 r_2^2 + r_1 r_3^2 + r_2^2 r_3 - r_2 r_3^2} - \frac{(a1\, r_2^2 - a1\, r_3^2 - a2\, r_1^2 + a2\, r_3^2 + a3\, r_1^2 - a3\, r_2^2)\, x}{(r_1 - r_2)\,(r_1 r_2 - r_1 r_3 - r_2 r_3 + r_3^2)} + \frac{a1\, r_2^2 r_3 - a1\, r_2 r_3^2 - a2\, r_1^2 r_3 + a2\, r_1 r_3^2 + a3\, r_1^2 r_2 - a3\, r_1 r_2^2}{(r_1 - r_2)\,(r_1 r_2 - r_1 r_3 - r_2 r_3 + r_3^2)} \qquad \textbf{(6.6)}$$

Avoir supposé que $Q \neq P$, c'est avoir supposé que $Q - P \neq 0$. Montrons que cette supposition nous conduit à se contredire. Calculons $Q - P$.

```
> 'Q'-'P'=Q-P;
```

$$Q - P = \frac{(a1\, r_2 - a1\, r_3 - a2\, r_1 + a2\, r_3 + a3\, r_1 - a3\, r_2)\, x^2}{r_1^2 r_2 - r_1^2 r_3 - r_1 r_2^2 + r_1 r_3^2 + r_2^2 r_3 - r_2 r_3^2} - \frac{(a1\, r_2^2 - a1\, r_3^2 - a2\, r_1^2 + a2\, r_3^2 + a3\, r_1^2 - a3\, r_2^2)\, x}{(r_1 - r_2)\,(r_1 r_2 - r_1 r_3 - r_2 r_3 + r_3^2)} + \frac{a1\, r_2^2 r_3 - a1\, r_2 r_3^2 - a2\, r_1^2 r_3 + a2\, r_1 r_3^2 + a3\, r_1^2 r_2 - a3\, r_1 r_2^2}{(r_1 - r_2)\,(r_1 r_2 - r_1 r_3 - r_2 r_3 + r_3^2)} - \frac{a1\,(x - r_2)\,(x - r_3)}{(r_1 - r_2)\,(r_1 - r_3)} - \frac{a2\,(x - r_1)\,(x - r_3)}{(r_2 - r_1)\,(r_2 - r_3)} - \frac{a3\,(x - r_1)\,(x - r_2)}{(r_3 - r_1)\,(r_3 - r_2)} \qquad \textbf{(6.7)}$$

Simplifions $Q - P$.

```
> ``=normal(rhs(%));
```

$$= 0 \qquad \textbf{(6.8)}$$

Notre hypothèse est contredite car $P = Q$.
Cela établit que le polynôme

$P = \dfrac{a_1\,(x - r_2)\,(x - r_3)}{(r_1 - r_2)\,(r_1 - r_3)} + \dfrac{a_2\,(x - r_1)\,(x - r_3)}{(r_2 - r_1)\,(r_2 - r_3)} + \dfrac{a_3\,(x - r_1)\,(x - r_2)}{(r_3 - r_1)\,(r_3 - r_2)}$ est l'unique polynôme quadratique tel que $P(r_1) = a_1$, $P(r_2) = a_2$, $P(r_3) = a_3$.

```
> unassign('a','b','c');
```